The Sine Gordon Equation

In[1]:= `Clear["Global`*"]`

The Equation

The equation is

$$\frac{\partial^2 u}{\partial x^2} - \frac{\partial^2 u}{\partial t^2} = \sin u$$

In[2]:=
```
sinegordoneq[u_] := D[u[x, t], {x, 2}] - D[u[x, t], {t, 2}] == Sin[u[x, t]] // FullSimplify
```

Without the right hand side we would have the wave equation with the wave speed, c, set to unity.

One Soliton Solutions

We display a soliton solution; first of all for the special case of no time dependence. We verify that the equation is satisfied by giving the command **sinegordoneq** and seeing that the result is **True**.

In[3]:=
```
u[x_, t_] = 4 ArcTan[Exp[x]]; u[x, t] // TraditionalForm
```

Out[3]//TraditionalForm=

$$4\tan^{-1}(e^x)$$

In[4]:= `sinegordoneq[u]`

Out[4]= True

From the plot below we see that u varies between 0 (for large negative x) and 2π (for large positive x).

In[5]:=
```
Plot[{u[x, 0], 2 π}, {x, -4, 4}, PlotRange → {0, 6.5}, PlotStyle →
  {{Hue[0], AbsoluteThickness[2]}, {Hue[0.7], AbsoluteThickness[2], Dashing[{0.02, 0.02}]}}]
```

Out[5]=



A more general solution with time dependence is

$$u(x, t) = 4\tan^{-1}\left[\exp\left[(x - vt)\Big/\sqrt{1 - v^2}\right]\right]$$

which shows that the soliton travels with velocity v . Note that $1/\sqrt{1-v^2}$ is a "Lorentz contraction" factor. We verify again that this satisfies the equation.

```
In[6]:= u[x_, t_] := 4 ArcTan[Exp[(x - v t) / Sqrt[1 - v^2]]]
```

```
In[7]:= sinegordoneq[u]
Out[7]= True
```

Now we choose v = 0.8 and produce a simple animation of the soliton. We integrate up to time t = 20, during which the soliton moves a distance x = 16.

```
In[8]:= v = 0.8;
```

```
In[9]:= Animate[Plot[u[x, t], {x, -5, 22}, PlotRange → {0, 6.5}], {t, 0, 20, 0.25}]
```

Out[9]=



▪ The "Breather" Solution

There are also multi-soliton solutions, which we don't show, and a "breather" solution, a bound soliton-anti soliton pair, which oscillates and stays close to the origin. Here it is:

```
In[10]:= ub[x_, t_] = 4 ArcTan[Sin[t / Sqrt[2]] / Cosh[x / Sqrt[2]]]; ub[x, t] // TraditionalForm
```

Out[10]//TraditionalForm=

$$4\tan^{-1}\left(\text{sech}\left(\frac{x}{\sqrt{2}}\right)\sin\left(\frac{t}{\sqrt{2}}\right)\right)$$

We verify that it satisfies the sine Gordon equation

```
In[11]:= sinegordoneq[ub]
Out[11]= True
```

and show an animation of it (the period is $2\sqrt{2}\,\pi$). Note that the largest range of u is at x=0 where u varies between $-\pi$ and π .

In[12]:=
```
Animate[Plot[ub[x, t], {x, -6, 6}, PlotRange → {-3.2, 3.2}],
 {t, 0, N[Sqrt[2] 2 Pi], N[Sqrt[2] 2 Pi] / 80}]
```

Out[12]=